

Health Technology Assessment PRO Least Squares Mean Difference Programming Techniques Using SAS®

Yirong Cao MSD, Hoddesdon, U.K.
Sory Traore MSD, Hoddesdon, U.K.

ABSTRACT

Comparison of Patient Reported Outcome (PRO) scores is an essential part of Health Technology Assessment (HTA) analysis to exhibit the effect of treatment against control. Starting from a two sample t-test, this article provides a detailed description of how to compute least squares means and p-values from t-test, analysis of covariance (ANCOVA) model, longitudinal ANCOVA and constrained longitudinal data analysis (cLDA) by introducing covariates using PROC TTEST and PROC MIXED in SAS. This article uses a simulated data sample to reflect the real world data of randomised clinical trials (RCTs). A brief discussion of simulation is also provided.

KEY WORDS: PRO, HTA, ANCOVA, longitudinal ANCOVA, cLDA, least squares means

I. INTRODUCTION

PRO is an essential part of Health Technology Assessment (HTA) analyses. A widely accepted standardised PRO measure is the EQ5D questionnaire, with which patients report the degree of health status via visual analogue scale (VAS) scores or time trade off (TTO) scores. The treatment effect can be exhibited through comparing mean difference of scores between treatment and control arms. Besides the commonly used arithmetic mean, least squares mean is a more efficient way to compare the group mean values. The illustration provided contains a simulated sample of 400 subjects with 4 visits. The SAS procedures on how to calculate least squares mean difference using analysis of covariance (ANCOVA) and longitudinal ANCOVA and constrained longitudinal data analysis (cLDA) with PROC MIXED procedures are shown step by step. This paper aims to review and clarify the similarities and differences between approaches commonly used to analyse change from baseline and to provide efficient ways to fit the chosen model in SAS.

The structure is shown as below: Section II describes the simulation of the sample data and produces arithmetic means with two sample t-tests using PROC TTEST; Section III illustrates how to calculate LS means and p-values with ANCOVA, longitudinal ANCOVA and cLDA models respectively through PROC MIXED; Section IV draws the conclusions.

II. DATA SIMULATION AND ARITHMETIC MEAN

2.1 SIMULATION BACKGROUND

To illustrate the differences among the various standard approaches to analyse change from baseline of PRO, we simulated data that exhibit the most common scenarios in randomised clinical trials.

A sample of 400 subjects was drawn from a multivariate normal distribution of 4 visit repeated measurements (Y_0, Y_1, Y_2, Y_3) for each subject, including one baseline Y_0 measurement and three post-treatment measurements Y_1, Y_2 and Y_3 .

PhUSE 2017

Let $\mu^{\text{Control}} = (\mu_0, \mu_4, \mu_8, \mu_{12})$ be the theoretical mean vector in the control arm at visits $t=0, 4, 8,$ and 12 . Assuming a linear time effect on response measurement over time $4, 8$ and 12 , $\mu_t = \mu_0 - t$, the mean vector in the control arm was numerically set to $\mu^{\text{Control}} = (80, 76, 72, 68)$.

Let $\alpha = (0, \alpha_4, \alpha_8, \alpha_{12})$ be the vector of treatment effect. The vector of means of responses in the treatment arm at visit $t=0, 4, 8,$ and 12 is $\mu^{\text{treatment}} = (\mu_0, \mu_4 + \alpha_4, \mu_8 + \alpha_8, \mu_{12} + \alpha_{12})$. The treatment effect was assumed to be linear over time $\alpha_t = \frac{1}{2}t$, then the mean vector in the treatment arm was numerically set to $\mu^{\text{treatment}} = (80, 78, 76, 74)$. The variance-covariance matrix Σ was assumed to be the same for control and treatment (homoscedasticity assumption).

The components Y_0, Y_1, Y_2 and Y_3 were assumed to be correlated and their variances are assumed to be homogeneous, $\sigma^2 = 256$. Measurements in a same subject are assumed to be correlated following a first order autoregressive AR(1) structure with $\rho=0.7$ (this implies that neighbour measurements have the same correlation).

The variance-covariance matrix is as follows:

$$\Sigma = \sigma^2 \begin{bmatrix} 1 & \rho & \rho^2 & \rho^3 \\ \rho & 1 & \rho & \rho^2 \\ \rho^2 & \rho & 1 & \rho \\ \rho^3 & \rho^2 & \rho & 1 \end{bmatrix} \text{ This numerically equals } \Sigma = \begin{bmatrix} 256 & 179 & 125 & 88 \\ 179 & 256 & 179 & 125 \\ 125 & 179 & 256 & 179 \\ 88 & 125 & 179 & 256 \end{bmatrix}$$

Both intermittent missing values based on missing completely at random (MCAR) mechanism and monotone pattern based on missing at random (MAR) mechanism were introduced as follows.

Let R_{it} be the indicator of missingness at time t . If the measurement Y_{it} of the response at each time t for the subject i is missing, then $R_{it} = 1$. If this value is observed, then $R_{it} = 0$.

The baseline measurement was assumed to be MCAR with probability of missingness $\Pr(R_{i0} = 1) = 0.05$. The post-treatment measurement was assumed to include both mild intermittent missing values based MCAR with probability of missing $P(R_{it}) = 0.025$ for each time t and monotone missing values based on MAR that was generated using a logistic regression model (Diggle, 1993 and Lu, 2010):

$$\text{logit}(\Pr(R_{it}|R_{it-1} = 1)) = -3.5 + 0.05(100 - Y_{it-1})$$

The simulation codes are presented in Appendix 1. Here is a screenshot of the simulated data:

	subjid	trtp	aval	trtpn	base	chg	visit
1	1	Treatment	81.343003268	1	81.343003268	0	0
2	1	Treatment	73.481768728	1	81.343003268	-7.86123454	4
3	1	Treatment	64.685570717	1	81.343003268	-16.65743255	8
4	1	Treatment	.	1	81.343003268	.	12
5	2	Treatment	57.462834304	1	57.462834304	0	0
6	2	Treatment	49.187706966	1	57.462834304	-8.275127337	4
7	2	Treatment	50.287745138	1	57.462834304	-7.175089166	8
8	2	Treatment	.	1	57.462834304	.	12
9	3	Treatment	79.650007895	1	79.650007895	0	0
10	3	Treatment	82.666939662	1	79.650007895	3.0169317674	4
11	3	Treatment	84.577653716	1	79.650007895	4.9276458216	8
12	3	Treatment	87.304174565	1	79.650007895	7.6541666705	12
13	4	Treatment	71.212507818	1	71.212507818	0	0
14	4	Treatment	84.019672685	1	71.212507818	12.807164867	4
15	4	Treatment	.	1	71.212507818	.	8
16	4	Treatment	.	1	71.212507818	.	12
17	5	Treatment	94.502424966	1	94.502424966	0	0
18	5	Treatment	99.687910161	1	94.502424966	5.1854851953	4
19	5	Treatment	93.30537393	1	94.502424966	-1.197051036	8
20	5	Treatment	87.348558641	1	94.502424966	-7.153866324	12
21	6	Treatment	80.596015463	1	80.596015463	0	0
22	6	Treatment	87.476978227	1	80.596015463	6.8809627637	4
23	6	Treatment	.	1	80.596015463	.	8
24	6	Treatment	.	1	80.596015463	.	12
25	7	Treatment	79.585382082	1	79.585382082	0	0

PhUSE 2017

SUBJID identifies the subjects, TRTPN and TRTP are the numeric and character codes for the treatment (1: for the active treatment and 2: for control). BASE is the baseline measurement for the outcome of interest. AVAL represents the post-baseline measurements of the outcome of interest; VISIT is the numeric value characterizing the week to which the record of AVAL belongs. CHG represents change from baseline.

2.2 T TEST

A two-sample t-test comparing arithmetic means is a common method to distinguish score difference between treatment and control. This method is straightforward and the results are easy to explain. Arithmetic means can be obtained through PROC TTEST with the following codes. The results of the t-tests and confidence intervals are saved by the ODS output statement.

```
proc ttest data=analdata alpha=0.05;
  where visit>0;
  by visit;
  class trtpn;
  var chg;
  ods output Conflimits=cl TTests=ttest;
run;
```

Table 1 shows the mean change from baseline at each time point and the corresponding difference between treatment and control. The t-test based p-value is statistically significant only at the last visit (week 12).

Table 1

Mean change from baseline and difference mean change from baseline at each visit based on T test

	Treatment Mean (95% CI)	Control Mean (95% CI)	Mean Difference	P-value
Week 4	-0.7 (-2.5,1.2)	-2.7 (-4.6,-0.7)	2.0 (-0.7,4.7)	0.1431
Week 8	-3.5 (-6.0,-0.9)	-7.0 (-9.6,-4.3)	3.5 (-0.2,7.2)	0.0618
Week 12	-6.1 (-9.2,-3.1)	-11.9 (-15.2,-8.7)	5.8 (1.3,10.3)	0.0111

III. ANCOVA MODEL

3.1 ANCOVA MODEL

Brief motivation of using ANCOVA over the previous unadjusted comparisons of mean change from baseline:

The classical approach to analyse change from baseline in clinical trials with a baseline measurement and one single post-baseline measurement is ANCOVA. With this model the treatment effect is assessed using the change from baseline as a dependent variable and adjusted on baseline measurement used as a covariate. It is well known in the statistical literature that ANCOVA is more appropriate than the unadjusted comparisons as it improves both the accuracy and efficiency of estimates and test statistics. Specifically it accounts for the eventual imbalance of treatment groups at baseline when the baseline and post-baseline measurements are correlated (Liu et al., 2009; Wei and Zhang, 1992; Frison, et al. 1989; Crager, 1987).

This section discusses how to produce least squares mean difference with an ANCOVA model using PROC MIXED. In this model, baseline is included as a covariate. The syntax is shown as below:

```
proc mixed data=analdata;
  where visit ^= 0;
  by visit;

  class trtpn;
  model chg = base trtpn /noint cl ddfm=KR;

  lsmeans trtpn / pdiff cl;
```

PhUSE 2017

```
ods output LSMeans=lsmean;
ods output diffs=diff;
run;
```

This code produces three models at different time points by using the 'by visit' statement. The dependent variable is the change between each time point and the baseline (chg). The predictive variables are baseline values (base) and treatment (trtpn). The ESTIMATE statement defines the difference as treatment LS means minus control LS means with the option of 'trtpn 1 -1'. This is because treatment (trtpn=1) has a smaller numerical code than control (trtpn=2). The results are shown as below:

Table 2

Mean change from baseline and difference mean change from baseline at each visit based on ANCOVA

	Treatment LS means (95% CI)	Control LS means (95% CI)	LS mean difference (95% CI)	P-value
Week 4	-0.8 (-2.5,1.0)	-2.5 (-4.3,-0.8)	1.7 (-0.7,4.2)	0.1649
Week 8	-3.7 (-5.8,-1.5)	-6.7 (-9.0,-4.5)	3.0 (-0.1,6.2)	0.0580
Week 12	-6.6 (-9.1,-4.1)	-11.3 (-14.1,-8.6)	4.7 (1.0,8.4)	0.0130

3.2 LONGITUDINAL ANCOVA MODEL

Brief motivation of using Longitudinal ANCOVA over the classical ANCOVA to analyse mean change from baseline:

It is very common that the outcome variable includes repeated post-baseline measurements; therefore a change is considered at each time point. These outcomes are analysed similarly to the classical one single post-baseline ANCOVA, thus the model includes the baseline measurement as a covariate alongside the treatment variable as a factor. This model is known as the longitudinal ANCOVA.

In addition to the correlation between baseline and post-baseline measurements that the classical ANCOVA accounts for, the longitudinal ANCOVA takes into account the correlations among post-baseline measurements. The SAS procedure PROC MIXED suitably fits the longitudinal ANCOVA model.

The longitudinal ANCOVA is preferable over the classical ANCOVA even in the case where the focus is on the last post-baseline measurement as it is more efficient in that it uses all the available data including the drop-out subjects. It can be represented with the following model:

$$E(Y_{ijt} | Y_{ij0}) = \alpha_t Y_{ij0} + \beta_{jt} I(\text{treatment} = j, \text{time} = t) + \gamma_t X_i$$

where Y_{ijt} is the PRO score for subject i , with treatment assignment j , at visit t which is equal to $1, 2, \dots, T$ and Y_{ij0} is the corresponding baseline measurement. α_t is the slope (effect) of baseline measurement at time t which is for both treatment groups. β_{jt} is the mean change from baseline for treatment group j at time t , and $I(\cdot)$ is the indicator function. X_i is the vector of stratification factors (binary) for this patient i and γ_t is the coefficient vector for stratification factors at time t . The model can be produced with the following SAS code:

```
proc sql;
select mean(aval) into: baselineMean
from analdata where visit=0;
quit;

proc mixed data=analdata;
where visit>0;
class visit trtpn subjid;
model chg = visit trtpn trtpn*visit base*visit /noint cl ddfm=KR;
repeated visit / subject=subjid type=UN;

estimate 'Treatment at Week 4' visit 1 0 0 trtpn 1 0 visit*trtpn 1 0 0 0 0 0
base*visit &baselineMean. 0 0/cl;
estimate 'Control at Week 4' visit 1 0 0 trtpn 0 1 visit*trtpn 0 1 0 0 0 0
```

PhUSE 2017

```

base*visit &baselineMean. 0 0/cl;
estimate 'Treatment versus control comparison at Week 4' trtpn 1 -1
visit*trtpn 1 -1 0 0 0 0/cl;

estimate 'Treatment at Week 8' visit 0 1 0 trtpn 1 0 visit*trtpn 0 0 1 0 0 0
base*visit 0 &baselineMean. 0/cl;
estimate 'Control at Week 8' visit 0 1 0 trtpn 0 1 visit*trtpn 0 0 0 1 0 0
base*visit 0 &baselineMean. 0/cl;
estimate 'Treatment versus control comparison at Week 8' trtpn 1 -1
visit*trtpn 0 0 1 -1 0 0/cl;

estimate 'Treatment at Week 12' visit 0 0 1 trtpn 1 0 visit*trtpn 0 0 0 0 1 0
base*visit 0 0 &baselineMean./cl;
estimate 'Control at Week 12' visit 0 0 1 trtpn 0 1 visit*trtpn 0 0 0 0 0 1
base*visit 0 0 &baselineMean./cl;
estimate 'Treatment versus control comparison at Week 12' trtpn 1 -1
visit*trtpn 0 0 0 0 1 -1/cl;

ODS OUTPUT estimates=la_est;
run;

```

The results of mean difference from longitudinal models show a similar profile to the ANCOVA model, and p-values are significant at week 8 and week 12.

Table 3

Mean change from baseline and difference mean change from baseline at each visit based on Longitudinal ANCOVA

	Treatment LS means (95% CI)	Control LS means (95% CI)	LS mean difference (95% CI)	P-value
Week 4	-0.4 (-2.1,1.4)	-2.1 (-3.8,-0.3)	1.7 (-0.8,4.2)	0.1766
Week 8	-2.4 (-4.6,-0.3)	-6.1 (-8.3,-3.9)	3.7 (0.6,6.7)	0.0199
Week 12	-5.0 (-7.5,-2.6)	-10.2 (-12.9,-7.6)	5.2 (1.6,8.8)	0.0046

3.3 cLDA MODEL

Brief motivation of using cLDA over Longitudinal ANCOVA to analyse mean change from baseline:

The constrained longitudinal data analysis (cLDA) is a model that includes the baseline measurement as a dependant variable unlike the longitudinal ANCOVA where the baseline measurement is included as a covariate. However, the means of baseline measurement in the control and treatment arms are constrained to be the same as by definition that there is no treatment effect at baseline after randomisation. This model proposed by Liang and Zeger (2000) is shown to overcome some of the limitations of the longitudinal ANCOVA (Liu et al., 2009; Lu, 2010). Indeed, cLDA requires weaker assumptions for the missing data mechanism than the longitudinal ANCOVA. The within treatment group point estimates from cLDA are unbiased under the missing at random (MAR) mechanism, whereas in longitudinal ANCOVA, subjects with missing baseline or without any post-baseline measurement are excluded, therefore the resulting point estimates are based on a completers analysis which requires a missing completely at random (MCAR) mechanism to be unbiased. Moreover, it is also shown that the estimated variance of treatment effect is smaller using cLDA model than the longitudinal ANCOVA (Liu et al., 2009; Lu, 2010). The cLDA model is specified as follows:

$$E(Y_{ijt}) = \beta_0 + \beta_{jt}I(t > 0) + \gamma_t X_i$$

where Y_{ijt} is the PRO score for subject i , with treatment assignment j , at visit t , β_0 is the baseline mean for all treatment groups, β_{jt} is the mean change from baseline for treatment group j at time t , and $I(\cdot)$ is the indicator function. X_i is the stratification factor (binary) vector for this patient i , and γ_t is the coefficient vector for stratification factors at time t .

There are three steps to produce LS means with cLDA models using the SAS PROC MIXED procedure:

PhUSE 2017

- 1) Create macro variables for the treatment number, the number of visits and a string of time points. The statement of 'select...into:' is used to create macro variables, and strip function removes the trailing and back spaces out of the macro variable values.

```
proc sql;
  select distinct trtpn into: treatno
    from analdata
    where trtp='Treatment';

  select count(distinct visit) into: visitn
    from analdata;

  select distinct visit into: time_points separated by ' '
    from analdata;
quit;

%let cvisitn=%sysfunc(strip(&visitn));
%let cvisitn1=%sysfunc(strip(%eval(&visitn-1)));

%put Treatment number: >>>- &treatno -<<<;
%put Visit numbers: >>>- &visitn -<<<;
%put Time points: >>>- &time_points-<<<;
%put Value of cvisitn: >>>- &cvisitn -<<<;
%put Value of cvisitn1: >>>- &cvisitn1 -<<<;
```

The values of the macro values are shown as below:

```
491 %put Treatment number: >>>-&treatno-<<<;
Treatment number: >>>- 1-<<<
492 %put Visit numbers: >>>-&visitn-<<<;
Visit numbers: >>>- 4-<<<
493 %put Time points: >>>-&time_points-<<<;
Time points: >>>-0 4 8 12-<<<
494 %put Value of cvisitn: >>>-&cvisitn-<<<;
Value of cvisitn: >>>-4-<<<
495 %put Value of cvisitn1: >>>-&cvisitn1-<<<;
Value of cvisitn1: >>>-3-<<<
```

- 2) Create dummy variables for modelling. Three array statements are used to create a number of variables: vst(&cvisitn) produces four variables (0, 4, 8 and 12); t(&cvisitn) array produces dummy variables, e.g. if visit is equal to vst4(i.e. 12), then t3 is 1, otherwise t3 is 0. Tta(&cvisitn) array creates dummy variables for the interaction of time and visit: e.g., if t3 is equal to 1 and trtpn is 1 (treatment group), then tta3 is set to 1; otherwise tta3 is 0.

```
data analdata;
  set analdata;

  if visit in (&time_points);

  array vst(&cvisitn) (&time_points);
  array t(&cvisitn) t0-t&cvisitn1;
  array tta(&cvisitn) tta0-tta&cvisitn1;

  do i = 1 to &visitn;
    t(i) = (visit = vst(i));
    tta(i) = t(i) * (trtpn = &treatno);
  end;
  drop i;
run;
```

- 3) Implement the mixed model:

```
proc mixed data=analdata order=internal;
  class subjid visit;
  model aval =visit tta1 tta2 tta3 / noint solution cl ddfm = kr
  outpred=pred_output;
  repeated visit / subject = subjid type = un;

  estimate 'Treatment at Week 4' visit -1 1 0 0 tta1 1/cl;
  estimate 'Control at Week 4' visit -1 1 0 0/cl;
  estimate 'Treatment vs control comparison at Week 4' tta1 1/cl;
```

PhUSE 2017

```

estimate 'Treatment at Week 8' visit -1 0 1 0 tta2 1/cl;
estimate 'Control at Week 8' visit -1 0 1 0/cl;
estimate 'Treatment vs control comparison at Week 8' tta2 1/cl;

estimate 'Treatment at Week 12' visit -1 0 0 1 tta3 1/cl;
estimate 'Control at Week 12' visit -1 0 0 1/cl;
estimate 'Treatment vs control comparison at Week 12' tta3 1/cl;

ods output Estimates = est0;
run;

```

The above codes are used to compute the least squares mean difference for both treatments at all three time points (week 4, 8 and 12). The number -1 in the ESTIMATE statement represents minus the baseline value, 0 means that period is omitted, and 1 means the computed time points. The dummy variable such as “TTA1 1” is the treatment minus baseline. For control, this can be omitted. For example, the contrast (-1 0 0 1) represents the difference between week 12 and baseline. In the statements where the contrast is not shown, e.g. the last estimate statement, the procedure calculates the mean difference between the treatment and the control at week 12 by using ‘tta3 1’. Here are the results from the above code:

Table 4

Mean change from baseline and difference mean change from baseline at each visit based on cLDA

	Treatment LS means (95% CI)	Control LS means (95% CI)	LS mean difference (95% CI)	P-value
Week 4	-0.1 (-1.7,1.5)	-1.8 (-3.5,-0.2)	1.7 (-0.5,3.9)	0.1240
Week 8	-1.7 (-3.8,0.3)	-5.6 (-7.8,-3.4)	3.9 (1.0,6.7)	0.0073
Week 12	-4.0 (-6.4,-1.6)	-9.6 (-12.2,-7.0)	5.6 (2.2,9.0)	0.0011

IV. CONCLUSION

This paper introduces how to compare the mean difference of PRO scores with two sample t-tests using PROC TTEST and least squares means with p-value computed from classic ANCOVA, longitudinal ANCOVA and cLDA models with the PROC MIXED procedure in SAS. Each method has its own pros and cons. T-tests are the most straightforward, but may not be efficient in some cases. The classic ANCOVA model incorporates covariates into model fitting, and does not account for the correlations among repeated measures. Longitudinal ANCOVA and cLDA methods mitigate this drawback, and the latter is more efficient when the dataset contains missing values.

	SAS Procedures	Advantages	Disadvantages	Efficiency
t test	PROC TTEST	<ul style="list-style-type: none"> • Straightforward and robust 	<ul style="list-style-type: none"> • Least efficient 	+
ANCOVA	PROC MIXED	<ul style="list-style-type: none"> • Adjustment for covariate 	<ul style="list-style-type: none"> • Does not consider time series impact • Require data to be missing completely at random (completers analysis) 	++
Longitudinal ANCOVA	PROC MIXED	<ul style="list-style-type: none"> • Adjustment for covariate • Accounts for the correlation among repeated measures 	<ul style="list-style-type: none"> • Exclude subjects without baseline measurement or without any post-baseline measurement. • Requires repeated measures 	++
cLDA	PROC MIXED	<ul style="list-style-type: none"> • Adjustment for covariate • Accounts for the correlation among repeated measures • More efficient with data missing at random (MAR) • Use data from patients with missing baseline 	<ul style="list-style-type: none"> • Requires repeated measures 	+++

REFERENCES:

Crager, M. R. (1987). Analysis of Covariance in Parallel-group Clinical Trials with Pretreatment Baselines. *Biometrics* 43, 895–901.

Diggle, P. J. (1993). On informative and random dropouts in longitudinal studies (letter to editor). *Biometrics* 49, 947–949.

EQ-5D. <http://www.euroqol.org/about-eq-5d/how-to-use-eq-5d.html>

Frison, L. and Pocock, S. J. (1992). Repeated Measures in Clinical Trials: Analysis Using Mean Summary Statistics and Its Implications for Design. *Statistics in Medicine* 11, 1685–1704.

Liang, K. Y. and Zeger, S. (2000). Longitudinal Data Analysis of Continuous and Discrete Responses for Pre-post Designs. *Sankhya*: The Indian Journal of Statistics, Series B 62, 134–148.

Liu, G. F., Lu K., Mogg, R., Mallick, M. and Mehrotra, D. V. (2009) Should Baseline be a Covariate or Dependent Variable in Analyses of Change from Baseline in Clinical Trials? *Statistics in Medicine*, 28: 2509-2530.

Lu, K. (2010) On Efficiency of Constrained Longitudinal Data Analysis Versus Longitudinal Analysis of Covariance *Biometrics* 66, 891–896

SAS Institute Inc. SAS/STAT 9.3 User's Guide: The Mixed Procedure.

Zhang, W. (2006). Using the RAND Function in SAS for Data Simulation in Clinical Trials. SAS SUGI 31 Proceedings.

CONTACT INFORMATION:

Your comments and questions are valued and encouraged. Please contact:

Yirong Cao
MSD
Hoddesdon, Hertfordshire
EN11 9BU
United Kingdom
Yirong.cao@merck.com

Sory Traore
MSD
Hoddesdon, Hertfordshire
EN11 9BU
United Kingdom
Sory.traore@merck.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

APPENDIX: Simulation SAS Codes

1. Assumption of the statistical models and the simulation

Let's consider a subject $i = 1 \dots N$ in a study, Y_{it} a sequence of measurements on subject i at $(K + 1)$ time points, $t = 0, \dots, K$. Let $Y_i = (Y_{i0}, \dots, Y_{iK})$ be the corresponding outcome vector.

Let X_i equal (X_{i1}, \dots, X_{ip}) , a known vector of p components for each subject i for which the values can be either continuous variables or dummy variables indicating group membership and associated $\beta = (\beta_1, \dots, \beta_p)$ an unknown vector of parameters for fixed-effects; $Z_i = (Z_{i1}, \dots, Z_{ip})$

Similarly a known vector of q components Z_i for each subject i and an associated unknown vector γ for random-effect are defined. A mixed-effect model is generally formulated as follows:

$$Y_i = X_i\beta + Z_i\gamma + \varepsilon_i$$

PhUSE 2017

where ε_i denotes a vector of random errors with p components normally distributed with mean $0_p = (0, \dots, 0)$ and covariance matrix R independent of γ which is also normally distributed with mean $0_q = (0, \dots, 0)$ and covariance matrix G . In a matrix formulation where the vectors for subjects are stacked together:

$$Y = X^T\beta + Z^T\gamma + \varepsilon$$

The response vector Y is a normally distributed mean $\mu = X^T\beta$ and covariance matrix $\Sigma = ZGZ^T + R$

$$Y \sim \text{Normal}(X^T\beta, ZGZ^T + R)$$

The special case presented here assumes no random-effect $Z = 0$ and the variance component is reduced to the fixed-effect part $\Sigma = R$ only.

2. SAS code

```
*****;
** Data Preparation **;
*****;

%*control group;
proc iml;
  %*population means;
  Mean = {80, 76, 72, 68};

  %*population covariances;
  Cov = {256 179 125 88,
         179 256 179 125,
         125 179 256 179,
         88 125 179 256};

  %*sample size;
  N = 200;

  call randseed(123);

  %*x is a 200 x 4 matrix;
  X = RandNormal(N, Mean, Cov);
  SampleMean = mean(X);
  SampleCov = cov(X);
  varNames = "Y0":"Y3";

  print SampleMean[colname=varNames],
        SampleCov[colname=varNames rowname=VarNames];

  %* write sample to SAS data set;
  create MVN_ctl from X[colname=varNames];
  append from X;
  close MVN_ctl;
quit;

%*treatment group;
proc iml;
  %*population means;
  Mean = {80, 78, 76, 74};

  %*population covariances;
  Cov = {256 179 125 88,
         179 256 179 125,
         125 179 256 179,
         88 125 179 256};

  %*sample size;
  N = 200;

  call randseed(123);

  %*x is a 200 x 4 matrix;
  X = RandNormal(N, Mean, Cov);
  SampleMean = mean(X);
  SampleCov = cov(X);
  varNames = "Y0":"Y3";

  print SampleMean[colname=varNames],
        SampleCov[colname=varNames rowname=VarNames];

  %* write sample to SAS data set;
  create MVN_trt from X[colname=varNames];
  append from X;
  close MVN_trt;
```

PhUSE 2017

```
quit;

data simdata0;

    /*deriving subject id;
    subjid+1;
    length trtp $15.;

    set MVN_trt(in=a) MVN_ctl(in=b);

    /*deriving treatment number;
    trtpn = a + 2 * b;
    if trtpn = 1 then trtp='Treatment';
    else trtp='Control';
run;

data simdatal(keep=subjid trtp trtpn Y: R:);
    set simdata0;

    /*probability of MCAR missingness for baseline;
    call streaminit(123456);
    p=0.05;

    array y y0-y3;
    array logit logit0-logit3;

    /*logit for baseline;
    logit0=log(p/(1-p));

    do i=2 to dim(logit);
        /*logit for post-baseline missingness based on MAR mechanism;
        logit(i) = -3.5 + 0.05*(100-y(i-1));
    end;

    array U U0-U3;
    array p_pred p_pred0-p_pred3;
    array R R0-R3;

    do i=1 to dim(p_pred);

        /*probabilities of missingness based on MAR mechanism for post-baseline and on MCAR
        for baseline;
        p_pred(i) = exp(logit(i))/(1 + exp(logit(i)));
        U(i) = rand("Uniform");

        /*matrix of status of missingness: 0: observed, 1: missing;
        R(i) = U(i) < p_pred(i);
    end;

    /* introducing monotone missingness;
    if R1 > R2 then R2=R1;
    if R2 > R3 then R3=R2;

    /*introducing a mild intermittent missingness for post-baseline based on MCAR;
    p_mcar = 0.025;
    array Umcar Umcar1-Umcar3;
    array mcarR mcarR1-mcarR3;
    do i=1 to dim(Umcar);
        Umcar(i) = rand("Uniform");
        mcarR(i) = Umcar(i) < p_mcar;
        R(i+1) = max(R(i+1), mcarR(i));
    end;

run;

/*see the estimate the probability of monotone MAR + intermittent MCAR missingness;
proc freq data=simdatal;
    table R: ;
run;

data simdata2;
    set simdatal;
    array y y0-y3;
    array R R0-R3;
    do i=1 to dim(y);
        /*replacing observed values by missing values using the missing value matrix*;
        if R(i)=1 then y(i) = .;
```

PhUSE 2017

```
end;
base=y0;
run;

%*change format dataset from wide format into long format;
data simdata3;
  set simdata2(in=_0 keep = subjid trtp trtpn y0 base rename=(y0=aval))
    simdata2(in=_1 keep = subjid trtp trtpn y1 base rename=(y1=aval))
    simdata2(in=_2 keep = subjid trtp trtpn y2 base rename=(y2=aval))
    simdata2(in=_3 keep = subjid trtp trtpn y3 base rename=(y3=aval));

  %*deriving change from baseline;
  chg = aval - base;
  %*deriving visit variable;
  visit= _0*0 + _1*4 + _2*8 + _3*12;

  if aval>100 then aval=100;
  if base>100 then base=100;
run;

proc sort data=simdata3 out=analdata;
  by subjid visit;
run;
```